
INTRODUCTION

The infinite monkey theorem states that a monkey hitting keys at random on a keyboard for an 
infinite amount of  time will almost surely type or create a particular chosen text, such as the 
complete works of  Shakespeare. Now, suppose a keyboard has only alphabetical characters, 26 
keys, and the word to be typed in “bacon”. Typing at random, the chance that the first letter 
typed is b is 1/26, as is he chance that the second letter typed is a, and so on. The events are 
independent, so the chance of  the first five letters matching bacon is (1/26)5, or 0.00000219. 

This program uses a random number generator, to design an Ada program to simulate the typing 
monkeys. 

PROGRAM STRUCTURE

Here’s the core skeleton of  the program:

! with ada.text_io; use ada.text_io;
! with ada.integer_text_io; use ada.integer_text_io;
! with ada.float_text_io; use ada.float_text_io;
! with ada.numerics.elementary_functions;
! use ada.numerics.elementary_functions;
! with ada.numerics.discrete_random;

! procedure monkeyTypes is

! begin

! end monkeyTypes;

Nothing really different from the previous Ada pracniques. Except for the fact that a new 
package “ada.numerics.discrete_random” has been added. Two subroutines will be used 
from this package:

 reset() - a procedure to start the generator in a unique state in each run
	 random() - generate a random number

More on this in a minute.

SUBTYPES

The great thing about Ada is the ability to create subtypes - basically constrained types. In this 
program, two subtypes will be created: one to represent the 3 letter word, another to represent 
the 26 indexes for the alphabet array.

     subtype word is string (1..3);    
     w : word;  



    subtype ranchar is integer range 1..26;

Now notice, that the new random number package has not been instantiated withe the use 
clause yet. That is because the subtype ranchar will be used to limit the random numbers 
generated. By overloading in this manner it is possible to create a package of  subprograms to 
calculate random values between 1 and 26.

    package random_char is new ada.numerics.discrete_random(ranchar);
    use random_char;

PROCEDURE typing

The next portion of  the code really just encapsulates the “typing monkey” simulation. It’s basic 
structure looks like this:

    procedure typingM(n : in integer; text : in out word) is
        -- variable declarations
    begin
        -- algorithm workings
    end typingM;

The procedure has two parameters: n, the number of  iterations, and text, the word to match. 
What variables are needed?

s something to hold the continuously generated 3-letter word

x a variable to store the randomly generated number 

alphabet a string to hold the 26 characters of  the alphabet

found a variable to identify whether the word has been generated

rng a generator for the random number generator

prob a variable to calculate the probability of  finding the word

These are declared in the following manner:

    s : word;
    x : ranchar;
    alphabet : string(1..26);
    found : integer;
    rng : generator;
    prob : float;
  



STEP 1: PRELIMINARY SETUP

The first step in the algorithm requires resetting the random number generator, setting found to 
0, and creating the alphabet look-up table.

    reset(rng);
    found := 0;
    alphabet(1..26) := "abcdefghijklmnopqrstuvwxyz";

Next, set random characters in the first two positions of  the word s.

    x := random(rng); 
    s(1) := alphabet(x); 
    x := random(rng); 
    s(2) := alphabet(x); 

STEP 2: THE LOOP

The loop does all the hard work. 

1. Generate the random number to decide the random letter for the third letter of  the word.
2. Check the randomly generated word against the word being searched for.

1. If  they are equivalent, set found to the index of  the loop, and exit.
2. If  they are not equivalent, shift the letters in s.

    for i in 1..n loop
        -- Place a random character in the 3rd position
        x := random(rng); 
        s(3) := alphabet(x);

        -- If the generated text equals the user-input word, exit.
        if s = text then
            found := i;
            exit;
        end if;
        -- Otherwise shift characters 2 and 3 to positions 1 and 2
            s(1) := s(2);
            s(2) := s(3);
    end loop; 

STEP 3: THE OUTPUT

Finally, either the number of  loop iterations was exhausted or the word was found. So print out 
the appropriate information to the user.

    if found /= 0 then



        put("The word "); put(text); put(" was found after");
        put(found); put(" iterations"); put_line(" ");
        put("Probability = ");
        prob := 1.0 / float(found); 
        put(prob, aft=>5, exp=>0);
    else
        put("The word "); put(text); put(" was not found");
    end if;

The put() statements could be put on separate lines (pardon the pun)... but it might be cleaner 
this way.

THE WHOLE PROCEDURE

-- Subprogram to running the "typing monkey" simulation 
procedure typingM(n : in integer; text : in out word) is
    s : word;
    x : ranchar;
    alphabet : string(1..26);
    found : integer;
    rng : generator;
    prob : float;
begin
    reset(rng);
    found := 0;
    -- Create the look-up table
    alphabet(1..26) := "abcdefghijklmnopqrstuvwxyz";

    -- Place random characters in the first two positions of the word.
    x := random(rng); 
    s(1) := alphabet(x); 
    x := random(rng); 
    s(2) := alphabet(x); 

    for i in 1..n loop
        -- Place a random character in the 3rd position
        x := random(rng); 
        s(3) := alphabet(x);

        -- If the generated text equals the user-input word, exit.
        if s = text then
            found := i;
            exit;
        end if;
        -- Otherwise shift characters 2 and 3 to positions 1 and 2                            
        s(1) := s(2);
        s(2) := s(3);
    end loop; 



   if found /= 0 then
        put("The word "); put(text); put(" was found after");
        put(found); put(" iterations"); put_line(" ");
        put("Probability = ");
        prob := 1.0 / float(found); 
        put(prob, aft=>5, exp=>0);
    else
        put("The word "); put(text); put(" was not found");
    end if;
end typingM;

THE MAIN PROGRAM

This really just gets the user input and calls the procedure.

begin    
    put_line("Enter a three letter word> ");
    get(w);
    -- Run the simulation with word w, and for 1000,000 iterations
    typingM(100000,w);
end monkeyTypes;

TESTING

Use the program to count how many characters have to be generated until the word “vox” 
appears. The chance of  the first three letters matching “vox” is (1/26)3, or 0.0000569. Here are 
some sample runs:

Enter a three letter word>
vox
The word vox was found after       2890 iterations
Probability =  0.00035

Enter a three letter word>
vox
The word vox was found after      22693 iterations
Probability =  0.00004

Enter a three letter word>
vox
The word vox was found after        673 iterations
Probability =  0.00149



THE WHOLE PROGRAM

with ada.text_io; use ada.text_io;
with ada.integer_text_io; use ada.integer_text_io;
with ada.float_text_io; use ada.float_text_io;
with ada.numerics.elementary_functions;
use ada.numerics.elementary_functions;
with ada.numerics.discrete_random;

procedure monkeyTypes is

    -- Create a small string of length 3
    subtype word is string (1..3);    
    w : word;  

    -- Create a subtype of integers with values of 1 to 26
    subtype ranchar is integer range 1..26;

    -- Create a package of subprograms to calculate random values
    -- between 1 and 26 using the generic package discrete_random
    package random_char is new ada.numerics.discrete_random(ranchar);
    use random_char;
    
    -- Procedure typingM goes here

begin    
    put_line("Enter a three letter word> ");
    get(w);
    -- Run the simulation with word w, and for 1000,000 iterations
    typingM(100000,w);
end monkeyTypes;


